

上海大学计算机学院

《汇编语言程序设计》报告

姓名 周鹏飞 学号 20121333 指导教师 _____

实验名称: 分类统计字符个数

一、实验任务

用分支程序设计与循环程序设计的方式实现字母、数字以及其他字符的分类统计。

二、实验内容

1. 编写程序，程序代码如下：

datas segment

```
line_max_length    db 0ffh
line               db 0, 100h dup(?)
letter_count       dw 0
digit_count        dw 0
other_count        dw 0

input              db 'input a line:$'
output_letter_count db 0dh,0ah,'letter count:$'
output_digit_count db 0dh,0ah,'digit count:$'
output_other_count db 0dh,0ah,'other count:$'
```

datas ends

stacks segment stack

```
db 100h dup(?)
```

stacks ends

codes segment

assume cs:codes,ds:datas,ss:stacks

main proc far

start:

```
push ds
mov ax,0h
push ax
```

```

mov ax,datas          ;初始化 ds
mov ds,ax
;输入一行字符提示
lea dx,input
mov ah,9
int 21h
;输入一行字符
lea dx,line_max_length
mov ah,10
int 21h

;对输入的一行字符进行统计
mov cl,line
xor ch,ch
mov si,1
cmp cx,0              ;判断字符串的长度是否为 0
jz break              ;若为 0, 跳转
s:
mov al,line[si]
cmp al,'0'             ;判断字符是否小于'0'
jb other              ;若小于'0', 则字符为其他字符
cmp al,'9'            ;判断字符是否大于'9'
ja letter_or_other    ;若大于'9',则字符为字母或其他字符
inc digit_count       ;否则, 字符为数字字符, 数字记录器加 1
jmp next              ;跳转
letter_or_other:
and al,11011111b      ;将字符转为大写字符
cmp al,'A'            ;判断字符是否大于'A'
jb other              ;若小于, 则字符是其他字符, 跳转
cmp al,'Z'            ;判断字符是否大于'Z'
ja other              ;若大于, 则字符为其他字符, 跳转
inc letter_count      ;否则, 字符为字母字符, 字母记录器加 1
jmp next
other:
inc other_count        ;字符为其他字符, 其他记录器加 1
next:
inc si
loop s
break:
;输出记录器的值
lea dx,output_letter_count
mov ah,9
int 21h
mov ax,letter_count

```

```

    call decimal

    lea dx,output_digit_count
    mov ah,9
    int 21h
    mov ax,digit_count
    call decimal

    lea dx,output_other_count
    mov ah,9
    int 21h
    mov ax,other_count
    call decimal
    ret

main endp

decimal proc near
    push ax
    push cx
    push dx
    push bx

    cmp ax,0
    jge no_negative
    ;如果整数为负数，先输出负号，然后求补转化成整数再输出相应的数字
    mov bx,ax
    mov dl,'-'
    mov ah,2
    int 21h
    neg bx ;求补
    mov ax,bx

no_negative:
    mov cx,0
    mov bx,10
de:
    xor dx,dx
    div bx
    push dx ;将整数的低位压栈，低位先入栈
    inc cx;计数器，记录整数的位数
    cmp ax,0
    jnz de;非零跳转，如果这个整数还没有除尽，就继续除
de1:

```

```

pop dx;将余数一位一位的输出，从高位往低位
add dl,30h;将数字转化为字符
mov ah,2 ;输出
int 21h
loop de1 ;循环输出

```

```

pop bx
pop dx
pop cx
pop ax

```

ret;将处理器回到该程序被调用的程序上，即相当于高级语言中的 return

```

decimal endp
codes ends

```


```

end main


```

2. 画出代码的流程图

主程序流程图：



子程序流程图：



3. 执行结果

```
F:\ASM>test0.exe
input a line:aaaaaaaaaaaaaaaa;;;;;;;;;;;;;111111111111
letter count:16
digit count:13
other count:13
F:\ASM>test0.exe
input a line:
letter count:0
digit count:0
other count:0
```